Question: This is my code:

```
RevolutionPlot3D[x^2, {x, 0, 1}, RevolutionAxis -> "X", 
 MeshFunctions -> {Boole[#2 == #^2] &}, Mesh -> {{1}}, 
 MeshStyle -> Directive[Red, Thickness[0.03]], PlotPoints -> 200, 
 Boxed -> False, Axes -> True, AxesOrigin -> {0, 0, 0}, 
 AxesStyle -> {Red, Green, Blue}]

```

[![enter image description here](https://i.stack.imgur.com/q3YCv.png)](https://i.stack.imgur.com/q3YCv.png)

But if I set `BoundaryStyle -> None`, then all `MeshStyle` will disappear. So how to just keep the `MeshStyle` but don't get that red circle `BoundaryStyle`?

Answer: 
You can *hide* the unwanted ring using a combination of options `Exclusions`, `ExclusionsStyle` and the `Method` sub-option `"BoundaryOffset" -> False`:

```
RevolutionPlot3D[x^2, {x, 0, 1},
 RevolutionAxis -> "X", 
 PlotPoints -> 100, 
 ImageSize -> 600,
 ViewPoint -> {0.75, -1.5, 3},
 Boxed -> False, Axes -> True, AxesOrigin -> {0, 0, 0}, 
 AxesStyle -> {Red, Green, Blue}, 
 MeshFunctions -> {Boole[#^2 == #2] &}, 
 Mesh -> {{1}}, 
 MeshStyle -> Directive[Red, Thickness[0.01]],
 Method -> {"BoundaryOffset" -> False}, 
 Exclusions -> {x == 1}, 
 ExclusionsStyle -> None] /.  Line[x_] :> Tube[x, .015]

```

[![enter image description here](https://i.stack.imgur.com/VlcbY.png)](https://i.stack.imgur.com/VlcbY.png)

Alternatively, the combination of options `Method -> {"BoundaryOffset" -> False}`, `MeshFunctions -> {#5 &}`, `Mesh -> {{0}}` and `BoundaryStyle -> None` gives the same picture.

And a variation on Craig Carter's approach using `ParametricPlot3D`

```
rp = RevolutionPlot3D[x^2, {x, 0, 1},
 RevolutionAxis -> "X", 
 Mesh -> None,
 PlotPoints -> 100, 
 ImageSize -> 600,
 ViewPoint -> {0.75, -1.5, 3},
 Boxed -> False, Axes -> True, AxesOrigin -> {0, 0, 0}, 
 AxesStyle -> {Red, Green, Blue}];

Show[rp, 
  ParametricPlot3D[{t, 0, t^2}, {t, 0, 1}, 
    PlotStyle -> Directive[Red, CapForm["Round"], Tube[.015]]]]

```

[![enter image description here](https://i.stack.imgur.com/5lb6h.png)](https://i.stack.imgur.com/5lb6h.png)

I you want to add the `x==1` ring with a different style, you can add a second `ParametricPlot3D` to `Show`:

```
Show[rp, 
  ParametricPlot3D[{t, 0, t^2}, {t, 0, 1}, 
    PlotStyle -> Directive[Red, CapForm["Round"], Tube[.015]]], 
  ParametricPlot3D[{1, Cos @ t, Sin @ t}, {t, 0, 2 Pi}, 
    PlotStyle -> Directive[Green, CapForm["Round"], Tube[.01]]]]

```

[![enter image description here](https://i.stack.imgur.com/zcBsh.png)](https://i.stack.imgur.com/zcBsh.png)
This is a hack, but a working hack:

```
rp = RevolutionPlot3D[x^2, {x, 0, 1}, RevolutionAxis -> "X", 
  PlotPoints -> 200, MeshStyle -> None, Boxed -> False, Axes -> True, 
  AxesOrigin -> {0, 0, 0}, AxesStyle -> {Red, Green, Blue}]

Show[rp, Graphics3D[
  {Red, Tube[BSplineCurve[{#, 0, #^2} & /@ Range[0, 1, .1]]],
   Blue, 
   Tube[BSplineCurve[{1, Cos[#], Sin[#]} & /@ Range[0, 2 Pi, Pi/24]]]
   }
  ]
 ]

```